

# AVR Assembler Help

Welcome to the ATMEL AVR Assembler.

- [What's New](#)
- [Known Issues](#)

Please select between the following Help items:

- [General information gives general information about the Assembler](#)
- [Assembler source gives a brief description of what a source file looks like](#)
- [Instruction mnemonics describes the AVR Instruction set](#)
  - [Arithmetic and Logic Instructions](#)
  - [Branch Instructions](#)
  - [Data Transfer Instructions](#)
  - [Bit and Bit-test Instructions](#)
- [Assembler directives gives a description of the directives](#)
- [Expressions describes how to make constant expressions](#)
  - [Expression operands](#)
  - [Expression operators](#)
  - [Functions in expressions](#)

The Assembler is supplied as a MS-DOS command line program that can be used stand-alone or automatically invoked by AVR Studio. A description of how to use the Command line Assembler is included in this help file

## Device specific instruction set summaries:

The actual instruction set varies between the devices. Use these links to verify the instruction set for the desired device.

- [AT90S1200](#)
- [AT90S2313](#)
- [AT90S2323 and AT90S2343](#)
- [AT90S2333 and AT90S4433](#)
- [AT90S4414 and AT90S8515](#)
- [AT90S4434 and AT90S8535](#)
- [AT90C8534](#)
- [ATtiny10, ATtiny11 and ATtiny12](#)
- [ATtiny15](#)
- [ATtiny22](#)
- [ATtiny26](#)
- [ATtiny28](#)
- [ATmega8/8515/8535](#)
- [ATmega16](#)
- [ATmega161](#)
- [ATmega162](#)
- [ATmega163](#)
- [ATmega169](#)
- [ATmega32](#)
- [ATmega323](#)
- [ATmega103](#)

- [ATmega64/128](#)

### **New for this release (avrasm v1.56):**

Added device directives for ATmega162, ATmega169, ATmega8515, ATmega8535, ATtiny26 and AT86RF401.

## **What's New**

### **AVR Assembler v1.56 date 30-apr-2002**

- Included device directives for ATmega162, ATmega169, ATmega8515, ATmega8535, ATtiny26, AT86RF401.
- Added \*.def.inc files for the above mentioned devices.

## **General information**

The Assembler translates assembly source code into object code. The generated object code can be used as input to a simulator such as the ATMEL AVR Simulator or an emulator such as the ATMEL AVR In-Circuit Emulator. The Assembler also generates a PROMable code which can be programmed directly into the program memory of an AVR microcontroller

The Assembler generates fixed code allocations, consequently no linking is necessary.

The instruction set of the AVR family of microcontrollers is only briefly described, refer to the AVR Data Book in order to get more detailed knowledge of the instruction set for the different microcontrollers.

## **Assembler source**

The Assembler works on source files containing instruction mnemonics, labels and directives. The instruction mnemonics and the directives often take operands.

Code lines should be limited to 120 characters.

Every input line can be preceded by a label, which is an alphanumeric string terminated by a colon. Labels are used as targets for jump and branch instructions and as variable names in Program memory and RAM.

An input line may take one of the four following forms:

```
[label:] directive [operands] [Comment]
[label:] instruction [operands] [Comment]
Comment
Empty line
```

A comment has the following form:

```
; [Text]
```

Items placed in braces are optional. The text between the comment-delimiter (;) and the end of line (EOL) is ignored by the Assembler. Labels, instructions and directives are described in more detail later.

## Examples:

```
label:  .EQU var1=100 ; Set var1 to 100 (Directive)
        .EQU var2=200 ; Set var2 to 200

test:   rjmp test    ; Infinite loop (Instruction)
        ; Pure comment line

        ; Another comment line
```

Note that there are no restrictions with respect to column placement of labels, directives, comments or instructions.

## Instruction mnemonics

The Assembler accepts mnemonic instructions from the instruction set. A summary of the instruction set mnemonics and their parameters is given here. For a detailed description of the Instruction set, refer to the AVR Data Book.

### Arithmetic and Logic Instructions

Mnemonic	Operands	Description	Operation	Flags	Cycles
<a href="#">ADD</a>	<a href="#">Rd,Rr</a>	Add without Carry	$Rd = Rd + Rr$	Z,C,N,V,H,S	1
<a href="#">ADC</a>	<a href="#">Rd,Rr</a>	Add with Carry	$Rd = Rd + Rr + C$	Z,C,N,V,H,S	1
<a href="#">ADIW</a>	<a href="#">Rd, K</a>	Add Immediate To Word	$Rd+1 : Rd, K$	Z,C,N,V,S	2
<a href="#">SUB</a>	<a href="#">Rd,Rr</a>	Subtract without Carry	$Rd = Rd - Rr$	Z,C,N,V,H,S	1
<a href="#">SUBI</a>	<a href="#">Rd,K8</a>	Subtract Immediate	$Rd = Rd - K8$	Z,C,N,V,H,S	1
<a href="#">SBC</a>	<a href="#">Rd,Rr</a>	Subtract with Carry	$Rd = Rd - Rr - C$	Z,C,N,V,H,S	1
<a href="#">SBCI</a>	<a href="#">Rd,K8</a>	Subtract with Carry Immediate	$Rd = Rd - K8 - C$	Z,C,N,V,H,S	1
<a href="#">AND</a>	<a href="#">Rd,Rr</a>	Logical AND	$Rd = Rd \cdot Rr$	Z,N,V,S	1
<a href="#">ANDI</a>	<a href="#">Rd,K8</a>	Logical AND with Immediate	$Rd = Rd \cdot K8$	Z,N,V,S	1
<a href="#">OR</a>	<a href="#">Rd,Rr</a>	Logical OR	$Rd = Rd \vee Rr$	Z,N,V,S	1
<a href="#">ORI</a>	<a href="#">Rd,K8</a>	Logical OR with Immediate	$Rd = Rd \vee K8$	Z,N,V,S	1
<a href="#">EOR</a>	<a href="#">Rd,Rr</a>	Logical Exclusive OR	$Rd = Rd \oplus Rr$	Z,N,V,S	1
<a href="#">COM</a>	<a href="#">Rd</a>	One's Complement	$Rd = \$FF - Rd$	Z,C,N,V,S	1
<a href="#">NEG</a>	<a href="#">Rd</a>	Two's Complement	$Rd = \$00 - Rd$	Z,C,N,V,H,S	1
<a href="#">SBR</a>	<a href="#">Rd,K8</a>	Set Bit(s) in Register	$Rd = Rd \vee K8$	Z,C,N,V,S	1
<a href="#">CBR</a>	<a href="#">Rd,K8</a>	Clear Bit(s) in Register	$Rd = Rd \cdot (\$FF - K8)$	Z,C,N,V,S	1
<a href="#">INC</a>	<a href="#">Rd</a>	Increment Register	$Rd = Rd + 1$	Z,N,V,S	1
<a href="#">DEC</a>	<a href="#">Rd</a>	Decrement Register	$Rd = Rd - 1$	Z,N,V,S	1
<a href="#">TST</a>	<a href="#">Rd</a>	Test for Zero or Negative	$Rd = Rd \cdot Rd$	Z,C,N,V,S	1
<a href="#">CLR</a>	<a href="#">Rd</a>	Clear Register	$Rd = 0$	Z,C,N,V,S	1
<a href="#">SER</a>	<a href="#">Rd</a>	Set Register	$Rd = \$FF$	None	1

<a href="#">ADIW</a>	<a href="#">RdI,K6</a>	Add Immediate to Word	$RdH:RdL = RdH:RdL + K6$	Z,C,N,V,S	2
<a href="#">SBIW</a>	<a href="#">RdI,K6</a>	Subtract Immediate from Word	$RdH:RdL = RdH:RdL - K6$	Z,C,N,V,S	2
<a href="#">MUL</a>	<a href="#">Rd,Rr</a>	Multiply Unsigned	$R1:R0 = Rd * Rr$	Z,C	2
<a href="#">MULS</a>	<a href="#">Rd,Rr</a>	Multiply Signed	$R1:R0 = Rd * Rr$	Z,C	2
<a href="#">MULSU</a>	<a href="#">Rd,Rr</a>	Multiply Signed with Unsigned	$R1:R0 = Rd * Rr$	Z,C	2
<a href="#">FMUL</a>	<a href="#">Rd,Rr</a>	Fractional Multiply Unsigned	$R1:R0 = (Rd * Rr) \ll 1$	Z,C	2
<a href="#">FMULS</a>	<a href="#">Rd,Rr</a>	Fractional Multiply Signed	$R1:R0 = (Rd * Rr) \ll 1$	Z,C	2
<a href="#">FMULSU</a>	<a href="#">Rd,Rr</a>	Fractional Multiply Signed with Unsigned	$R1:R0 = (Rd * Rr) \ll 1$	Z,C	2

## Branch Instructions

Mnemonic	Operands	Description	Operation	Flags	Cycles
<a href="#">RJMP</a>	<a href="#">k</a>	Relative Jump	$PC = PC + k + 1$	None	2
<a href="#">IJMP</a>	None	Indirect Jump to ( <a href="#">Z</a> )	$PC = Z$	None	2
<a href="#">EIJMP</a>	None	Extended Indirect Jump ( <a href="#">Z</a> )	$STACK = PC+1, PC(15:0) = Z, PC(21:16) = EIND$	None	2
<a href="#">JMP</a>	<a href="#">k</a>	Jump	$PC = k$	None	3
<a href="#">RCALL</a>	<a href="#">k</a>	Relative Call Subroutine	$STACK = PC+1, PC = PC + k + 1$	None	3/4*
<a href="#">ICALL</a>	None	Indirect Call to ( <a href="#">Z</a> )	$STACK = PC+1, PC = Z$	None	3/4*
<a href="#">EICALL</a>	None	Extended Indirect Call to ( <a href="#">Z</a> )	$STACK = PC+1, PC(15:0) = Z, PC(21:16) = EIND$	None	4*
<a href="#">CALL</a>	<a href="#">k</a>	Call Subroutine	$STACK = PC+2, PC = k$	None	4/5*
<a href="#">RET</a>	None	Subroutine Return	$PC = STACK$	None	4/5*
<a href="#">RETI</a>	None	Interrupt Return	$PC = STACK$	I	4/5*
<a href="#">CPSE</a>	<a href="#">Rd,Rr</a>	Compare, Skip if equal	if ( $Rd == Rr$ ) $PC = PC + 2$ or $3$	None	1/2/3
<a href="#">CP</a>	<a href="#">Rd,Rr</a>	Compare	$Rd - Rr$	Z,C,N,V,H,S	1
<a href="#">CPC</a>	<a href="#">Rd,Rr</a>	Compare with Carry	$Rd - Rr - C$	Z,C,N,V,H,S	1
<a href="#">CPI</a>	<a href="#">Rd,K8</a>	Compare with Immediate	$Rd - K$	Z,C,N,V,H,S	1
<a href="#">SBRC</a>	<a href="#">Rr,b</a>	Skip if bit in register cleared	if( $Rr(b)==0$ ) $PC = PC + 2$ or $3$	None	1/2/3
<a href="#">SBRS</a>	<a href="#">Rr,b</a>	Skip if bit in register set	if( $Rr(b)==1$ ) $PC = PC + 2$ or $3$	None	1/2/3
<a href="#">SBIC</a>	<a href="#">P,b</a>	Skip if bit in I/O register cleared	if( $I/O(P,b)==0$ ) $PC = PC + 2$ or $3$	None	1/2/3
<a href="#">SBIS</a>	<a href="#">P,b</a>	Skip if bit in I/O register set	if( $I/O(P,b)==1$ ) $PC = PC + 2$ or $3$	None	1/2/3
<a href="#">BRBC</a>	<a href="#">s,k</a>	Branch if Status flag cleared	if( $SREG(s)==0$ ) $PC = PC + k + 1$	None	1/2
<a href="#">BRBS</a>	<a href="#">s,k</a>	Branch if Status flag set	if( $SREG(s)==1$ ) $PC = PC + k + 1$	None	1/2
<a href="#">BREQ</a>	<a href="#">k</a>	Branch if equal	if( $Z==1$ ) $PC = PC + k + 1$	None	1/2
<a href="#">BRNE</a>	<a href="#">k</a>	Branch if not equal	if( $Z==0$ ) $PC = PC + k + 1$	None	1/2
<a href="#">BRCS</a>	<a href="#">k</a>	Branch if carry set	if( $C==1$ ) $PC = PC + k + 1$	None	1/2
<a href="#">BRCC</a>	<a href="#">k</a>	Branch if carry cleared	if( $C==0$ ) $PC = PC + k + 1$	None	1/2

<a href="#">BRSH</a>	<a href="#">k</a>	Branch if same or higher	if(C==0) PC = PC + k + 1	None	1/2
<a href="#">BRLO</a>	<a href="#">k</a>	Branch if lower	if(C==1) PC = PC + k + 1	None	1/2
<a href="#">BRMI</a>	<a href="#">k</a>	Branch if minus	if(N==1) PC = PC + k + 1	None	1/2
<a href="#">BRPL</a>	<a href="#">k</a>	Branch if plus	if(N==0) PC = PC + k + 1	None	1/2
<a href="#">BRGE</a>	<a href="#">k</a>	Branch if greater than or equal (signed)	if(S==0) PC = PC + k + 1	None	1/2
<a href="#">BRLT</a>	<a href="#">k</a>	Branch if less than (signed)	if(S==1) PC = PC + k + 1	None	1/2
<a href="#">BRHS</a>	<a href="#">k</a>	Branch if half carry flag set	if(H==1) PC = PC + k + 1	None	1/2
<a href="#">BRHC</a>	<a href="#">k</a>	Branch if half carry flag cleared	if(H==0) PC = PC + k + 1	None	1/2
<a href="#">BRTS</a>	<a href="#">k</a>	Branch if T flag set	if(T==1) PC = PC + k + 1	None	1/2
<a href="#">BRTC</a>	<a href="#">k</a>	Branch if T flag cleared	if(T==0) PC = PC + k + 1	None	1/2
<a href="#">BRVS</a>	<a href="#">k</a>	Branch if overflow flag set	if(V==1) PC = PC + k + 1	None	1/2
<a href="#">BRVC</a>	<a href="#">k</a>	Branch if overflow flag cleared	if(V==0) PC = PC + k + 1	None	1/2
<a href="#">BRIE</a>	<a href="#">k</a>	Branch if interrupt enabled	if(I==1) PC = PC + k + 1	None	1/2
<a href="#">BRID</a>	<a href="#">k</a>	Branch if interrupt disabled	if(I==0) PC = PC + k + 1	None	1/2

\* Cycle times for data memory accesses assume internal memory accesses, and are not valid for accesses through the external RAM interface. For the instructions CALL, ICALL, EICALL, RCALL, RET and RETI, add three cycles plus two cycles for each wait state in devices with up to 16 bit PC (128KB program memory). For devices with more than 128KB program memory, add five cycles plus three cycles for each wait state.

## Data Transfer Instructions

Mnemonic	Operands	Description	Operation	Flags	Cycles
<a href="#">MOV</a>	<a href="#">Rd,Rr</a>	Copy register	Rd = Rr	None	1
<a href="#">MOVW</a>	<a href="#">Rd,Rr</a>	Copy register pair	Rd+1:Rd = Rr+1:Rr, r,d even	None	1
<a href="#">LDI</a>	<a href="#">Rd,K8</a>	Load Immediate	Rd = K	None	1
<a href="#">LDS</a>	<a href="#">Rd,k</a>	Load Direct	Rd = (k)	None	2*
<a href="#">LD</a>	<a href="#">Rd,X</a>	Load Indirect	Rd = (X)	None	2*
<a href="#">LD</a>	<a href="#">Rd,X+</a>	Load Indirect and Post-Increment	Rd = (X), X=X+1	None	2*
<a href="#">LD</a>	<a href="#">Rd,-X</a>	Load Indirect and Pre-Decrement	X=X-1, Rd = (X)	None	2*
<a href="#">LD</a>	<a href="#">Rd,Y</a>	Load Indirect	Rd = (Y)	None	2*
<a href="#">LD</a>	<a href="#">Rd,Y+</a>	Load Indirect and Post-Increment	Rd = (Y), Y=Y+1	None	2*
<a href="#">LD</a>	<a href="#">Rd,-Y</a>	Load Indirect and Pre-Decrement	Y=Y-1, Rd = (Y)	None	2*
<a href="#">LDD</a>	<a href="#">Rd,Y+q</a>	Load Indirect with displacement	Rd = (Y+q)	None	2*
<a href="#">LD</a>	<a href="#">Rd,Z</a>	Load Indirect	Rd = (Z)	None	2*
<a href="#">LD</a>	<a href="#">Rd,Z+</a>	Load Indirect and Post-Increment	Rd = (Z), Z=Z+1	None	2*
<a href="#">LD</a>	<a href="#">Rd,-Z</a>	Load Indirect and Pre-Decrement	Z=Z-1, Rd = (Z)	None	2*
<a href="#">LDD</a>	<a href="#">Rd,Z+q</a>	Load Indirect with displacement	Rd = (Z+q)	None	2*
<a href="#">STS</a>	<a href="#">k,Rr</a>	Store Direct	(k) = Rr	None	2*
<a href="#">ST</a>	<a href="#">X,Rr</a>	Store Indirect	(X) = Rr	None	2*
<a href="#">ST</a>	<a href="#">X+,Rr</a>	Store Indirect and Post-Increment	(X) = Rr, X=X+1	None	2*

<a href="#">ST</a>	<a href="#">-X,Rr</a>	Store Indirect and Pre-Decrement	$X=X-1, (X)=Rr$	None	2*
<a href="#">ST</a>	<a href="#">Y,Rr</a>	Store Indirect	$(Y) = Rr$	None	2*
<a href="#">ST</a>	<a href="#">Y+,Rr</a>	Store Indirect and Post-Increment	$(Y) = Rr, Y=Y+1$	None	2
<a href="#">ST</a>	<a href="#">-Y,Rr</a>	Store Indirect and Pre-Decrement	$Y=Y-1, (Y) = Rr$	None	2
<a href="#">ST</a>	<a href="#">Y+q,Rr</a>	Store Indirect with displacement	$(Y+q) = Rr$	None	2
<a href="#">ST</a>	<a href="#">Z,Rr</a>	Store Indirect	$(Z) = Rr$	None	2
<a href="#">ST</a>	<a href="#">Z+,Rr</a>	Store Indirect and Post-Increment	$(Z) = Rr, Z=Z+1$	None	2
<a href="#">ST</a>	<a href="#">-Z,Rr</a>	Store Indirect and Pre-Decrement	$Z=Z-1, (Z) = Rr$	None	2
<a href="#">ST</a>	<a href="#">Z+q,Rr</a>	Store Indirect with displacement	$(Z+q) = Rr$	None	2
<a href="#">LPM</a>	None	Load Program Memory	$R0 = (Z)$	None	3
<a href="#">LPM</a>	<a href="#">Rd,Z</a>	Load Program Memory	$Rd = (Z)$	None	3
<a href="#">LPM</a>	<a href="#">Rd,Z+</a>	Load Program Memory and Post-Increment	$Rd = (Z), Z=Z+1$	None	3
<a href="#">ELPM</a>	None	Extended Load Program Memory	$R0 = (RAMPZ:Z)$	None	3
<a href="#">ELPM</a>	<a href="#">Rd,Z</a>	Extended Load Program Memory	$Rd = (RAMPZ:Z)$	None	3
<a href="#">ELPM</a>	<a href="#">Rd,Z+</a>	Extended Load Program Memory and Post Increment	$Rd = (RAMPZ:Z), Z = Z+1$	None	3
<a href="#">SPM</a>	None	Store Program Memory	$(Z) = R1:R0$	None	-
<a href="#">ESPM</a>	None	Extended Store Program Memory	$(RAMPZ:Z) = R1:R0$	None	-
<a href="#">IN</a>	<a href="#">Rd,P</a>	In Port	$Rd = P$	None	1
<a href="#">OUT</a>	<a href="#">P,Rr</a>	Out Port	$P = Rr$	None	1
<a href="#">PUSH</a>	<a href="#">Rr</a>	Push register on Stack	$STACK = Rr$	None	2
<a href="#">POP</a>	<a href="#">Rd</a>	Pop register from Stack	$Rd = STACK$	None	2

\* Cycle times for data memory accesses assume internal memory accesses and are not valid for accesses through the external RAM interface. For the LD, ST, LDD, STD, LDS, STS, PUSH and POP instructions, add one cycle plus one cycle for each wait state.

## Bit and Bit-test Instructions

Mnemonic	Operands	Description	Operation	Flags	Cycles
<a href="#">LSL</a>	<a href="#">Rd</a>	Logical shift left	$Rd(n+1)=Rd(n), Rd(0)=0, C=Rd(7)$	Z,C,N,V,H,S	1
<a href="#">LSR</a>	<a href="#">Rd</a>	Logical shift right	$Rd(n)=Rd(n+1), Rd(7)=0, C=Rd(0)$	Z,C,N,V,S	1
<a href="#">ROL</a>	<a href="#">Rd</a>	Rotate left through carry	$Rd(0)=C, Rd(n+1)=Rd(n), C=Rd(7)$	Z,C,N,V,H,S	1
<a href="#">ROR</a>	<a href="#">Rd</a>	Rotate right through carry	$Rd(7)=C, Rd(n)=Rd(n+1), C=Rd(0)$	Z,C,N,V,S	1
<a href="#">ASR</a>	<a href="#">Rd</a>	Arithmetic shift right	$Rd(n)=Rd(n+1), n=0,\dots,6$	Z,C,N,V,S	1
<a href="#">SWAP</a>	<a href="#">Rd</a>	Swap nibbles	$Rd(3..0) = Rd(7..4), Rd(7..4) = Rd(3..0)$	None	1
<a href="#">BSET</a>	<a href="#">s</a>	Set flag	$SREG(s) = 1$	SREG(s)	1
<a href="#">BCLR</a>	<a href="#">s</a>	Clear flag	$SREG(s) = 0$	SREG(s)	1
<a href="#">SBI</a>	<a href="#">P,b</a>	Set bit in I/O register	$I/O(P,b) = 1$	None	2
<a href="#">CBI</a>	<a href="#">P,b</a>	Clear bit in I/O register	$I/O(P,b) = 0$	None	2
<a href="#">BST</a>	<a href="#">Rr,b</a>	Bit store from register to T	$T = Rr(b)$	T	1
<a href="#">BLD</a>	<a href="#">Rd,b</a>	Bit load from register to T	$Rd(b) = T$	None	1

<a href="#">SEC</a>	None	Set carry flag	C = 1	C	1
<a href="#">CLC</a>	None	Clear carry flag	C = 0	C	1
<a href="#">SEN</a>	None	Set negative flag	N = 1	N	1
<a href="#">CLN</a>	None	Clear negative flag	N = 0	N	1
<a href="#">SEZ</a>	None	Set zero flag	Z = 1	Z	1
<a href="#">CLZ</a>	None	Clear zero flag	Z = 0	Z	1
<a href="#">SEI</a>	None	Set interrupt flag	I = 1	I	1
<a href="#">CLI</a>	None	Clear interrupt flag	I = 0	I	1
<a href="#">SES</a>	None	Set signed flag	S = 1	S	1
<a href="#">CLN</a>	None	Clear signed flag	S = 0	S	1
<a href="#">SEV</a>	None	Set overflow flag	V = 1	V	1
<a href="#">CLV</a>	None	Clear overflow flag	V = 0	V	1
<a href="#">SET</a>	None	Set T-flag	T = 1	T	1
<a href="#">CLT</a>	None	Clear T-flag	T = 0	T	1
<a href="#">SEH</a>	None	Set half carry flag	H = 1	H	1
<a href="#">CLH</a>	None	Clear half carry flag	H = 0	H	1
<a href="#">NOP</a>	None	No operation	None	None	1
<a href="#">SLEEP</a>	None	Sleep	See instruction manual	None	1
<a href="#">WDR</a>	None	Watchdog Reset	See instruction manual	None	1
<a href="#">BREAK</a>	None	Execution Break	See instruction manual	None	1

The Assembler is not case sensitive.

The operands have the following forms:

Rd: Destination (and source) register in the register file

Rr: Source register in the register file

b: Constant (0-7), can be a constant expression

s: Constant (0-7), can be a constant expression

P: Constant (0-31/63), can be a constant expression

K6; Constant (0-63), can be a constant expression

K8: Constant (0-255), can be a constant expression

k: Constant, value range depending on instruction. Can be a constant expression

q: Constant (0-63), can be a constant expression

Rdl: R24, R26, R28, R30. For ADIW and SBIW instructions

X,Y,Z: Indirect address registers (X=R27:R26, Y=R29:R28, Z=R31:R30)

## Assembler directives

The Assembler supports a number of directives. The directives are not translated directly into opcodes. Instead, they are used to adjust the location of the program in memory, define macros, initialize memory and so on. An overview of the directives is given in the following table.

Directive	Description
<a href="#">BYTE</a>	<a href="#">Reserve byte to a variable</a>
<a href="#">CSEG</a>	<a href="#">Code Segment</a>
<a href="#">CSEGSIZE</a>	<a href="#">Program memory size</a>
<a href="#">DB</a>	<a href="#">Define constant byte(s)</a>
<a href="#">DEF</a>	<a href="#">Define a symbolic name on a register</a>
<a href="#">DEVICE</a>	<a href="#">Define which device to assemble for</a>
<a href="#">DSEG</a>	<a href="#">Data Segment</a>
<a href="#">DW</a>	<a href="#">Define Constant word(s)</a>
<a href="#">ENDM, ENDMACRO</a>	<a href="#">End macro</a>
<a href="#">EQU</a>	<a href="#">Set a symbol equal to an expression</a>
<a href="#">ESEG</a>	<a href="#">EEPROM Segment</a>
<a href="#">EXIT</a>	<a href="#">Exit from file</a>
<a href="#">INCLUDE</a>	<a href="#">Read source from another file</a>
<a href="#">LIST</a>	<a href="#">Turn listfile generation on</a>
<a href="#">LISTMAC</a>	<a href="#">Turn Macro expansion in list file on</a>
<a href="#">NOLIST</a>	<a href="#">Turn listfile generation off</a>
<a href="#">ORG</a>	<a href="#">Set program origin</a>
<a href="#">SET</a>	<a href="#">Set a symbol to an expression</a>

Note that all directives must be preceded by a period.

## BYTE - Reserve bytes to a variable

The BYTE directive reserves memory resources in the SRAM. In order to be able to refer to the reserved location, the BYTE directive should be preceded by a label. The directive takes one parameter, which is the number of bytes to reserve. The directive can only be used within a Data Segment (see directives CSEG and DSEG). Note that a parameter must be given. The allocated bytes are not initialized.

### Syntax:

```
LABEL: .BYTE expression
```

### Example:

```
.DSEG
var1:  .BYTE 1           ; reserve 1 byte to var1
table: .BYTE tab_size   ; reserve tab_size bytes

.CSEG
    ldi r30,low(var1)   ; Load Z register low
    ldi r31,high(var1)  ; Load Z register high
    ld r1,Z             ; Load VAR1 into register 1
```

## CSEG - Code segment

The CSEG directive defines the start of a Code Segment. An Assembler file can consist of several Code Segments, which are concatenated into one Code Segment when assembled. The BYTE directive can not be used within a Code Segment. The default segment type is Code. The Code Segments have their own location counter which is a word counter. The ORG directive can be used to place code and constants at specific locations in the Program memory. The directive does not take any parameters.

### Syntax:



```
.CSEG
```

**Example:**

```
.DSEG                ; Start data segment
vartab: .BYTE 4       ; Reserve 4 bytes in SRAM

.CSEG                ; Start code segment
const: .DW 2          ; Write 0x0002 in prog.mem.
        mov r1,r0     ; Do something
```

**CSEGSIZE - Program Memory Size**

AT94K devices have a user configurable memory partition between the AVR Program memory and the data memory. The program and data SRAM is divided into three blocks: 10K x 16 dedicated program SRAM, 4K x 8 dedicated data SRAM, and 6K x 16 or 12K x 8 configurable SRAM which may be swapped between program and data memory spaces in 2K x 16 or 4K x 8 partitions.

This directive is used to specify the size of the program memory block.

**Syntax:**

```
.CSEGSIZE = 10 | 12 | 14 | 16
```

**Example:**

```
.CSEGSIZE = 12        ; Specifies the program meory size as 12K x 16
```

**DB - Define constant byte(s) in program memory and EEPROM**

The DB directive reserves memory resources in the program memory or the EEPROM memory. In order to be able to refer to the reserved locations, the DB directive should be preceded by a label. The DB directive takes a list of expressions, and must contain at least one expression. The DB directive must be placed in a Code Segment or an EEPROM Segment.

The expression list is a sequence of expressions, delimited by commas. Each expression must evaluate to a number between -128 and 255. If the expression evaluates to a negative number, the 8 bits twos complement of the number will be placed in the program memory or EEPROM memory location.

If the DB directive is given in a Code Segment and the expressionlist contains more than one expression, the expressions are packed so that two bytes are placed in each program memory word. *If the expressionlist contains an odd number of expressions, the last expression will be placed in a program memory word of its own, even if the next line in the assembly code contains a DB directive.* The unused half of the program word is set to zero. A warning is given, in order to notify the user that an extra zero byte is added to the .DB statement

**Syntax:**

```
LABEL: .DB expressionlist
```

**Example:**

```
.CSEG
consts: .DB 0, 255, 0b01010101, -128, 0xaa

.ESEG
```

```
const2: .DB 1,2,3
```

## DEF - Set a symbolic name on a register

The DEF directive allows the registers to be referred to through symbols. A defined symbol can be used in the rest of the program to refer to the register it is assigned to. A register can have several symbolic names attached to it. A symbol can be redefined later in the program.

### Syntax:

```
.DEF Symbol=Register
```

### Example:

```
.DEF temp=R16
.DEF ior=R0

.CSEG
ldi temp,0xf0 ; Load 0xf0 into temp register
in ior,0x3f ; Read SREG into ior register
eor temp,ior ; Exclusive or temp and ior
```

## DEVICE - Define which device to assemble for

The DEVICE directive allows the user to tell the Assembler which device the code is to be executed on. Using this directive, a warning is issued if an instruction not supported by the specified device occurs. If the Code Segment or EEPROM Segment are larger than supplied by the device, a warning message is given. If the directive is not used, it is assumed that all instructions are supported and that there are no restrictions on Program and EEPROM memory.

### Syntax:

```
.DEVICE <device code>
```

### Table: Device codes:

Classic	Tiny	Mega	Other
AT90S120	ATtiny11	ATmega8	AT94K
AT90S2313	ATtiny12	ATmega16	AT86RF401
AT90S2323	ATtiny22	ATmega161	
AT90S2333	ATtiny26	ATmega162	
AT90S4414		ATmega163	
AT90S4434		ATmega32	
AT90S8515		ATmega323	
AT90S8534		ATmega103	
AT90S8535		ATmega104	
AT90S2343		ATmega8515	
AT90S4433		ATmega8535	
		ATmega64	
		ATmega128	

### Example:

```
.DEVICE AT90S1200 ; Use the AT90S1200

.CSEG
```

```

push r30    ; This statement will generate a warning
            ; since the specified device does not
            ; have this instruction

```

Note: There has been a change of names that took effect 14.06.2001. The following devices are affected:

Old name	New name
ATmega104	ATmega128
ATmega32	ATmega323
ATmega164	ATmega16

In order NOT to break old projects, both old and new device directives are allowed for the parts that are affected.

## DSEG - Data Segment

The DSEG directive defines the start of a Data Segment. An Assembler file can consist of several Data Segments, which are concatenated into one Data Segment when assembled. A Data Segment will normally only consist of BYTE directives (and labels). The Data Segments have their own location counter which is a byte counter. The ORG directive can be used to place the variables at specific locations in the SRAM. The directive does not take any parameters.

### Syntax:

```
.DSEG
```

### Example:

```

.DSEG                ; Start data segment
var1: .BYTE 1        ; reserve 1 byte to var1
table: .BYTE tab_size ; reserve tab_size bytes.

.CSEG
    ldi r30,low(var1) ; Load Z register low
    ldi r31,high(var1) ; Load Z register high
    ld r1,Z           ; Load var1 into register 1

```

## DW - Define constant word(s) in program memory and EEPROM

The DW directive reserves memory resources in the program memory or the EEPROM memory. In order to be able to refer to the reserved locations, the DW directive should be preceded by a label.

The DW directive takes a list of expressions, and must contain at least one expression. The DB directive must be placed in a Code Segment or an EEPROM Segment.

The expression list is a sequence of expressions, delimited by commas. Each expression must evaluate to a number between -32768 and 65535. If the expression evaluates to a negative number, the 16 bits twos complement of the number will be placed in the program memory or EEPROM memory location.

### Syntax:

```
LABEL: .DW expressionlist
```

### Example:

```
.CSEG
varlist: .DW 0, 0xffff, 0b1001110001010101, -32768, 65535

.ESEG
eevarlst: .DW 0,0xffff,10
```

## ENDMACRO - End macro

The ENDMACRO directive defines the end of a Macro definition. The directive does not take any parameters. See the MACRO directive for more information on defining Macros.

### Syntax:

```
.ENDMACRO
```

### Example:

```
.MACRO SUBI16                ; Start macro definition
    subi r16,low(@0)        ; Subtract low byte
    sbci r17,high(@0)       ; Subtract high byte
.ENDMACRO
```

## EQU - Set a symbol equal to an expression

The EQU directive assigns a value to a label. This label can then be used in later expressions. A label assigned to a value by the EQU directive is a constant and can not be changed or redefined.

### Syntax:

```
.EQU label = expression
```

### Example:

```
.EQU io_offset = 0x23
.EQU porta     = io_offset + 2

.CSEG                ; Start code segment
    clr r2           ; Clear register 2
    out porta,r2     ; Write to Port A
```

## ESEG - EEPROM Segment

The ESEG directive defines the start of an EEPROM Segment. An Assembler file can consist of several EEPROM Segments, which are concatenated into one EEPROM Segment when assembled. An EEPROM Segment will normally only consist of DB and DW directives (and labels). The EEPROM Segments have their own location counter which is a byte counter. The ORG directive can be used to place the variables at specific locations in the EEPROM. The directive does not take any parameters.

### Syntax:

```
.ESEG
```

### Example:

```
.DSEG                ; Start data segment
var1: .BYTE 1        ; reserve 1 byte to var1
table: .BYTE tab_size ; reserve tab_size bytes.
```

```
.ESEG
eevar1: .DW 0xffff          ; initialize 1 word in EEPROM
```

## EXIT - Exit this file

The EXIT directive tells the Assembler to stop assembling the file. Normally, the Assembler runs until end of file (EOF). If an EXIT directive appears in an included file, the Assembler continues from the line following the INCLUDE directive in the file containing the INCLUDE directive.

### Syntax:

```
.EXIT
```

### Example:

```
.EXIT ; Exit this file
```

## INCLUDE - Include another file

The INCLUDE directive tells the Assembler to start reading from a specified file. The Assembler then assembles the specified file until end of file (EOF) or an EXIT directive is encountered. An included file may itself contain INCLUDE directives.

### Syntax:

```
.INCLUDE "filename"
```

### Example:

```
; iodefs.asm:
.EQU sreg    = 0x3f      ; Status register
.EQU sphigh  = 0x3e      ; Stack pointer high
.EQU splow   = 0x3d      ; Stack pointer low

; incdemo.asm
.INCLUDE iodefs.asm     ; Include I/O definitions
        in r0,sreg      ; Read status register
```

## LIST - Turn the listfile generation on

The LIST directive tells the Assembler to turn listfile generation on. The Assembler generates a listfile which is a combination of assembly source code, addresses and opcodes. Listfile generation is turned on by default. The directive can also be used together with the NOLIST directive in order to only generate listfile of selected parts of an assembly source file.

### Syntax:

```
.LIST
```

### Example:

```
.NOLIST          ; Disable listfile generation
.INCLUDE "macro.inc" ; The included files will not
.INCLUDE "const.def" ; be shown in the listfile
.LIST           ; Reenable listfile generation
```

## LISTMAC - Turn macro expansion on

The LISTMAC directive tells the Assembler that when a macro is called, the expansion of the macro is to be shown on the listfile generated by the Assembler. The default is that only the macro-call with parameters is shown in the listfile.

**Syntax:**

```
.LISTMAC
```

**Example:**

```
.MACRO MACX          ; Define an example macro
    add  r0,@0      ; Do something
    eor  r1,@1      ; Do something
.ENDMACRO           ; End macro definition

.LISTMAC            ; Enable macro expansion
    MACX r2,r1      ; Call macro, show expansion
```

**MACRO - Begin macro**

The MACRO directive tells the Assembler that this is the start of a Macro. The MACRO directive takes the Macro name as parameter. When the name of the Macro is written later in the program, the Macro definition is expanded at the place it was used. A Macro can take up to 10 parameters. These parameters are referred to as @0-@9 within the Macro definition. When issuing a Macro call, the parameters are given as a comma separated list. The Macro definition is terminated by an ENDMACRO directive.

By default, only the call to the Macro is shown on the listfile generated by the Assembler. In order to include the macro expansion in the listfile, a LISTMAC directive must be used. A macro is marked with a + in the opcode field of the listfile.

**Syntax:**

```
.MACRO macroname
```

**Example:**

```
.MACRO SUBI16          ; Start macro definition
    subi @1,low(@0)    ; Subtract low byte
    sbci @2,high(@0)   ; Subtract high byte
.ENDMACRO             ; End macro definition

.CSEG                ; Start code segment
    SUBI16 0x1234,r16,r17 ; Sub.0x1234 from r17:r16
```

**NOLIST - Turn listfile generation off**

The NOLIST directive tells the Assembler to turn listfile generation off. The Assembler normally generates a listfile which is a combination of assembly source code, addresses and opcodes. Listfile generation is turned on by default, but can be disabled by using this directive. The directive can also be used together with the LIST directive in order to only generate listfile of selected parts of an assembly source file.

**Syntax:**

```
.NOLIST
```

**Example:**

```
.NOLIST                ; Disable listfile generation
.INCLUDE "macro.inc"   ; The included files will not
.INCLUDE "const.def"   ; be shown in the listfile
.LIST                  ; Reenable listfile generation
```

**ORG - Set program origin**

The ORG directive sets the location counter to an absolute value. The value to set is given as a parameter. If an ORG directive is given within a Data Segment, then it is the SRAM location counter which is set, if the directive is given within a Code Segment, then it is the Program memory counter which is set and if the directive is given within an EEPROM Segment, it is the EEPROM location counter which is set. If the directive is preceded by a label (on the same source code line), the label will be given the value of the parameter. The default values of the Code and the EEPROM location counters are zero, and the default value of the SRAM location counter is 32 (due to the registers occupying addresses 0-31) when the assembling is started. Note that the SRAM and EEPROM location counters count bytes whereas the Program memory location counter counts words.

**Syntax:**

```
.ORG expression
```

**Example:**

```
.DSEG                ; Start data segment

.ORG 0x37            ; Set SRAM address to hex 37
variable: .BYTE 1    ; Reserve a byte at SRAM adr.37H

.CSEG
.ORG 0x10            ; Set Program Counter to hex 10
    mov r0,r1        ; Do something
```

**SET - Set a symbol equal to an expression**

The SET directive assigns a value to a label. This label can then be used in later expressions. A label assigned to a value by the SET directive can be changed later in the program.

**Syntax:**

```
.SET label = expression
```

**Example:**

```
.SET io_offset = 0x23
.SET porta      = io_offset + 2

.CSEG
    clr r2      ; Clear register 2
    out porta,r2 ; Write to Port A
```

**Expressions**

The Assembler incorporates expressions. Expressions can consist of [operands](#), [operators](#) and

[functions](#). All expressions are internally 32 bits.

## Operands

The following operands can be used:

- User defined labels which are given the value of the location counter at the place they appear.
- User defined variables defined by the SET directive
- User defined constants defined by the EQU directive
- Integer constants: constants can be given in several formats, including
  - Decimal (default): 10, 255
  - Hexadecimal (two notations): 0x0a, \$0a, 0xff, \$ff
  - Binary: 0b00001010, 0b11111111
  - Octal (leading zero): 010, 077
- PC - the current value of the Program memory location counter

## Operators

The Assembler supports a number of operators which are described here. The higher the precedence, the higher the priority. Expressions may be enclosed in parentheses, and such expressions are always evaluated before combined with anything outside the parentheses.

The following operators are defined:

Symbol	Description
!	<a href="#">Logical Not</a>
~	<a href="#">Bitwise Not</a>
-	<a href="#">Unary Minus</a>
*	<a href="#">Multiplication</a>
/	<a href="#">Division</a>
+	<a href="#">Addition</a>
-	<a href="#">Subtraction</a>
<<	<a href="#">Shift left</a>
>>	<a href="#">Shift right</a>
<	<a href="#">Less than</a>
<=	<a href="#">Less than or equal</a>
>	<a href="#">Greater than</a>
>=	<a href="#">Greater than or equal</a>
==	<a href="#">Equal</a>
!=	<a href="#">Not equal</a>
&	<a href="#">Bitwise And</a>
^	<a href="#">Bitwise Xor</a>
	<a href="#">Bitwise Or</a>
&&	<a href="#">Logical And</a>



<a href="#">11</a>	<a href="#">Logical Or</a>
--------------------	----------------------------

## Logical Not

Symbol: `!`  
Description: Unary operator which returns 1 if the expression was zero, and returns 0 if the expression was nonzero  
Precedence: 14  
Example: `ldi r16,!0xf0 ; Load r16 with 0x00`

## Bitwise Not

Symbol: `~`  
Description: Unary operator which returns the input expression with all bits inverted  
Precedence: 14  
Example: `ldi r16,~0xf0 ; Load r16 with 0x0f`

## Unary Minus

Symbol: `-`  
Description: Unary operator which returns the arithmetic negation of an expression  
Precedence: 14  
Example: `ldi r16,-2 ; Load -2(0xfe) in r16`

## Multiplication

Symbol: `*`  
Description: Binary operator which returns the product of two expressions  
Precedence: 13  
Example: `ldi r30,label*2 ; Load r30 with label*2`

## Division

Symbol: `/`  
Description: Binary operator which returns the integer quotient of the left expression divided by the right expression  
Precedence: 13  
Example: `ldi r30,label/2 ; Load r30 with label/2`

## Addition

Symbol: `+`  
Description: Binary operator which returns the sum of two expressions  
Precedence: 12  
Example: `ldi r30,c1+c2 ; Load r30 with c1+c2`

## Subtraction

Symbol: `-`  
Description: Binary operator which returns the left expression minus the right expression

Precedence: 12  
Example: `ldi r17,c1-c2 ;Load r17 with c1-c2`

### Shift left

Symbol: `<<`  
Description: Binary operator which returns the left expression shifted left the number given by the right expression  
Precedence: 11  
Example: `ldi r17,1<<bitmask ;Load r17 with 1 shifted left bitmask times`

### Shift right

Symbol: `>>`  
Description: Binary operator which returns the left expression shifted right the number given by the right expression  
Precedence: 11  
Example: `ldi r17,c1>>c2 ;Load r17 with c1 shifted right c2 times`

### Less than

Symbol: `<`  
Description: Binary operator which returns 1 if the signed expression to the left is Less than the signed expression to the right, 0 otherwise  
Precedence: 10  
Example: `ori r18,bitmask*(c1<c2)+1 ;Or r18 with an expression`

### Less or equal

Symbol: `<=`  
Description: Binary operator which returns 1 if the signed expression to the left is Less than or Equal to the signed expression to the right, 0 otherwise  
Precedence: 10  
Example: `ori r18,bitmask*(c1<=c2)+1 ;Or r18 with an expression`

### Greater than

Symbol: `>`  
Description: Binary operator which returns 1 if the signed expression to the left is Greater than the signed expression to the right, 0 otherwise  
Precedence: 10  
Example: `ori r18,bitmask*(c1>c2)+1 ;Or r18 with an expression`

### Greater or equal

Symbol: `>=`  
Description: Binary operator which returns 1 if the signed expression to the left is Greater than or Equal to the signed expression to the right, 0 otherwise  
Precedence: 10  
Example: `ori r18,bitmask*(c1>=c2)+1 ;Or r18 with an expression`

### Equal

Symbol: ==  
Description: Binary operator which returns 1 if the signed expression to the left is Equal to the signed expression to the right, 0 otherwise  
Precedence: 9  
Example: `andi r19,bitmask*(c1==c2)+1 ;And r19 with an expression`

### Not equal

Symbol: !=  
Description: Binary operator which returns 1 if the signed expression to the left is Not Equal to the signed expression to the right, 0 otherwise  
Precedence: 9  
Example: `.SET flag=(c1!=c2) ;Set flag to 1 or 0`

### Bitwise And

Symbol: &  
Description: Binary operator which returns the bitwise And between two expressions  
Precedence: 8  
Example: `ldi r18,High(c1&c2) ;Load r18 with an expression`

### Bitwise Xor

Symbol: ^  
Description: Binary operator which returns the bitwise Exclusive Or between two expressions  
Precedence: 7  
Example: `ldi r18,Low(c1^c2) ;Load r18 with an expression`

### Bitwise Or

Symbol: |  
Description: Binary operator which returns the bitwise Or between two expressions  
Precedence: 6  
Example: `ldi r18,Low(c1|c2) ;Load r18 with an expression`

### Logical And

Symbol: &&  
Description: Binary operator which returns 1 if the expressions are both nonzero, 0 otherwise  
Precedence: 5  
Example: `ldi r18,Low(c1&& c2) ;Load r18 with an expression`

### Logical Or

Symbol: ||  
Description: Binary operator which returns 1 if one or both of the expressions are nonzero, 0 otherwise  
Precedence: 4  
Example: `ldi r18,Low(c1||c2) ;Load r18 with an expression`

## Functions

The following functions are defined:

- LOW(expression) returns the low byte of an expression
- HIGH(expression) returns the second byte of an expression
- BYTE2(expression) is the same function as HIGH
- BYTE3(expression) returns the third byte of an expression
- BYTE4(expression) returns the fourth byte of an expression
- LWRD(expression) returns bits 0-15 of an expression
- HWRD(expression) returns bits 16-31 of an expression
- PAGE(expression) returns bits 16-21 of an expression
- EXP2(expression) returns 2 to the power of expression
- LOG2(expression) returns the integer part of  $\log_2(\text{expression})$